



DECUS

PROGRAM LIBRARY

DECUS NO.	8-130A
TITLE	REBIL8 - RELOCATING BINARY LOADER
AUTHOR	R. F. LaFontaine
COMPANY	Division of Mechanical Engineering C.S.I.R.O. Melbourne, Australia
DATE	March 4, 1968
SOURCE LANGUAGE	

DECUS

PROGRAM LIBRARY



DECUS NO.

FILE

AUTHOR

COMPANY

DATE

SOURCE/PRICE/NOTES

REBIL8 - RELOCATING BINARY LOADER

DECUS Program Library Write-up

DECUS No. 8-130A

1. Abstract.

Sections of the DEC-08-LBAA-LA binary loader have been re-written to extend its duties to loading of suitably prepared relocatable binary program tapes. Core storage requirements for the revised loader are the same as the standard DEC loader.

2. Requirements.

2.1 Storage.

The program uses 93 storage locations.
7612-7616; 7626-7753; 7776-7777

2.2 Equipment.

Minimum- PDP8/S and ASR-33

3. Usage.

3.1 Loading.

The program is loaded by the RIM loader. See digital 8-1-U for the loading procedure.

3.2 Switch Settings.

The switch register is used to input address modification when loading relocatable binary tapes. Note: REBIL8 does not use the switch register to indicate reader selection.

3.3 Start up.

Place the binary tape to be loaded in the reader and switch the reader ON. At installations having two readers, switch the second reader OFF. Switching the reader OFF LINE or removing tape from the reader will suffice. Place 7777 in the switch register and press the load address key.

LOADING STANDARD DEC PROGRAM TAPES

Press the start key and loading commences. The computer halts at the end of program reading with the checksum difference displayed in the accumulator. A satisfactory load is indicated when AC=0. See Section 3.5 for loading errors.

LOADING RELOCATABLE PROGRAMS

Place the address modification (the number to be added to each address) in the switch register. Press the start key and loading commences. Refer Section 5.2 for a discussion on address modification.

3.5 Errors.

Check that only one reader is ON during load time.
Reload the program by repositioning the tape, then press the continue key.

4. Restrictions.

4.1 Programs not relocatable.

Programs originally assembled in memory page 0 are not relocatable, as the page bit in memory reference instructions is 0.

Data located specifically in page 0 for direct reference by other pages are not relocatable.

4.2 Precautions.

Care should be taken when indirectly addressing data or program relocated at page 0 in locations 10-17, as these locations are the auto-index registers.

Programs relocated at page 0 must make proper use of memory locations 0 and 1 if the program is to operate with interrupt ON.

The loader must read a minimum of one tape leader code or the memory field may not be correctly defined.

5. Description.

5.1 Discussion.

The duties of this loader are similar to those of the DEC-08-LBAA-LA loader. However, when loading relocatable program tapes, two additional duties are imposed; address modification and data modification.

5.2 Address Modification.

The value contained in the switch register at start time is added to addresses read from the binary tape, using two's complement arithmetic, to produce new absolute addresses.

The content of the switch register is masked so that bits 0-4 only are used. Thus addresses can only be modified in units of 200 octal memory locations.

The following example shows the effect of switch register setting on addresses.

ADDRESS READ FROM TAPE	SR SETTING	FINAL ADDRESS
1250	0200	1450
0067	1453	1467
6000	4000	2000
3500	0000	3500
3500	7777	3500

5.3 Data Modification.

When a program is relocated, it is necessary to modify data which are used by memory reference instructions for indirect addressing. As an example, the following program is to be relocated at address 600.

0200	1221	TAD ALPHA	
0201	4620	JMS I LOC	
0202	0405	PARA	/data to be modified
.	.	.	
0220	1200 LOC,	1200	/data to be modified
0221	0407 ALPHA,	0407	
.	.	.	

After relocation, the program appears in the memory as :

0600	1221
0601	4620
0602	1005
.	.
0620	1600
0621	0407
.	.

Data intended for modification is marked on the binary tape by a 376 code. This code is placed behind the two 6-bit characters from which the data is assembled.

5.4 Relocatable Tapes

The program RELCON converts DEC binary tapes to relocatable tapes. RELCON can adjust addresses found on the DEC tapes so that the lowest recorded address will appear on the relocatable tape within the range 0-177.

In most cases, this simplifies Switch Register setting. As an example; assume a DEC binary program starts at address 400. RELCON changes the addresses so that the starting address for the relocatable version is at location 0000.

To load this program so that it starts at 1200, the loading procedure would be :

7777 is placed in the Switch Register and the Load Address key pressed. 1200 is placed in the Switch Register and the Start key is pressed. The program is now loaded into memory, and at completion of loading the program is started at the address indicated by the Switch Register.

5.5 Relocating Unconverted DEC Tapes.

Single page programs are relocatable provided that they do not contain data to be used by memory reference instructions for indirect addressing. This restriction does not apply to data locations that are filled with return addresses by the JMS instruction.

The procedure for relocating these tapes is shown in the example below.

7777 is placed in the Switch Register, then the Load Address key is pressed. If the lowest address on the tape is 7400 and it is required to relocate the program at 5400, place 6000 (-2000) in the Switch Register. Press start and the program commences loading.

Note: In the example above, the program starting address and the value contained in the Switch Register will probably bear no relation to each other.

6. Format.

The format of a relocatable tape is similar to the format of a DEC binary tape. The only difference is the inclusion of the 376 data modification mark. The following example shows a binary tape containing data (data=4437) and the 376 code.

```
      .  
      .  
00100.100      (44)  
00011.111      (37)  
11111.110      (376)  
      .  
      .
```



```

/ REBIL8
/ RELOCATING BINARY LOADER FOR THE PDP8/S
*7612
7612 0000 CHKSUM, 0
7613 0000 ADRMOD, 0
7614 0000 HIBYTE, 0
7615 0000 LOBYTE, 0
7616 0000 DFIELD, 0
*7626
7626 0000 INPUT, 0 /EXTRACTS DIAGNOSTICS, FIELD, L/T,
7627 3376 DCA STORE /AND THE DATA MODIFICATION MARK.
7630 7240 STA /PICK UP -1
7631 1376 TAD STORE /THE AC NOW CONTAINS -1 OR 375
7632 3352 DCA FLAG /SET DATA MOD. FLAG
7633 4254 JMS READ /INPUT A CHARACTER
7634 5240 JMP .+4 /FOUND L/T, DATA, OR DATA MOD. MARK
7635 4254 JMS READ /FOUND DIAGNOSTIC
7636 5235 JMP .-1 /LOOP UNTIL END OF DIAGNOSTIC
7637 5233 JMP .-4 /FOUND END OF DIAGNOSTIC
7640 1246 TAD M200
7641 7510 SPA /TEST FOR MEMORY DATA OR ADDRESS
7642 2226 ISZ INPUT /FOUND DATA OR AN ADDRESS
7643 1265 TAD M100
7644 7500 SMA /TEST FOR FIELD
7645 5251 JMP .+4 /FOUND FIELD
7646 7600 M200, -200 /CLEAR ACCUMULATOR
7647 5626 JMP I INPUT /EXIT FOR DATA, ADDRESS, OR L/T
7650 6214 RDF /PICK UP FIELD
7651 1351 TAD CDFINS /PICK UP CDF INSTRUCTION
7652 3216 DCA DFIELD /PLACE IN TEMPORARY STORAGE
7653 5233 JMP INPUT+5
7654 0000 READ, 0 /INPUTS A CHARACTER FROM THE L.S.
7655 6011 RSF /OR THE H.S. READER
7656 5271 JMP LSREAD /READER NOT READY; TRY L.S.READER
7657 6016 RRB RFC
7660 3376 DCA STORE
7661 1376 TAD STORE
7662 1307 TAD M376
7663 7450 SNA /TEST FOR DATA MOD MARK (376)
7664 5230 JMP INPUT+2 /FOUND DATA MOD MARK
7665 7700 M100, SMA CLA /TEST FOR DIANOSTIC MARK (377)
7666 2254 ISZ READ /FOUND 377; INC. RETURN ADDRESS
7667 1376 TAD STORE
7670 5654 JMP I READ
7671 6031 LSREAD, KSF
7672 5255 JMP READ+1 /READER NOT READY; TRY H.S.READER
7673 6036 KRB
7674 5260 JMP READ+4

```

7675	0000	ASEMB,	0	/WORD ASSEMBLER
7676	1214		TAD HIBYTE	
7677	7106		CLL RTL	
7700	7006		RTL	
7701	7006		RTL	/FIRST CHARACTER NOW IN AC0-AC5
7702	1215		TAD LOBYTE	/ADD SECOND CHARACTER
7703	5675		JMP I ASEMB	/EXIT; L=1=ADDRESS, L=0=DATA
7704	4275	FINIS,	JMS ASEMB	/ASSEMBLE TAPE CHECKSUM
7705	7041		CIA	
7706	1212		TAD CHKSUM	
7707	7402	M376,	HLT	
7710	6032	BEGIN,	KCC	
7711	6014		RFC	
7712	7604		LAS	/PICK UP ADDRESS MODIFICATION
7713	7001		IAC	/CONVERT TO 0000 IF SR WAS 7777
7714	0246		AND M200	/MASK ADDRESS MOD.
7715	3213		DCA ADRMOD	
7716	4226		JMS INPUT	
7717	5250		JMP M200+2	/SEARCH FOR END OF TAPE LEADER
7720	3212	BACK,	DCA CHKSUM	
7721	1216		TAD DFIELD	
7722	3344		DCA DFINST	/SET UP DATA FIELD INSTRUCTION
7723	1376		TAD STORE	
7724	3214		DCA HIBYTE	
7725	4254		JMS READ	
7726	3215		DCA LOBYTE	
7727	4226		JMS INPUT	
7730	5304		JMP FINIS	/FOUND TAPE TRAILER
7731	4275		JMS ASEMB	
7732	7420		SNL	/TEST FOR DATA OR ADDRESS
7733	5342		JMP DATA	
7734	1213		TAD ADRMOD	/FOUND ADDRESS; ADD ADDRESS MOD
7735	3353		DCA ADRES	
7736	1212	BACK2,	TAD CHKSUM	
7737	1215		TAD LOBYTE	
7740	1214		TAD HIBYTE	
7741	5320		JMP BACK	/JUMP BACK AND UPDATE CHECKSUM
7742	2352	DATA,	ISZ FLAG	/TEST FOR DATA MODIFICATION
7743	1213		TAD ADRMOD	/ADD ADDRESS MODIFICATION
7744	0000	DFINST,	0	/SELECT FIELD
7745	3753		DCA I ADRES	/STORE THE DATA
7746	2353		ISZ ADRES	/UPDATE ADDRESS
7747	1352		TAD FLAG	/PICK UP 376 IF DATA MODIFIED
7750	5336		JMP BACK2	
7751	6201	CDFINS,	CDF	
7752	0000	FLAG,	0	
7753	0000	ADRES,	0	
		*7776		
7776	0000	STORE,	0	
7777	5310		JMP BEGIN	

ADRES	7753
ADRMOD	7613
ASEMB	7675
BACK	7720
BACK2	7736
BEGIN	7710
CDFINS	7751
CHKSUM	7612
DATA	7742
DFIELD	7616
DFINST	7744
FINIS	7704
FLAG	7752
HIBYTE	7614
INPUT	7626
LOBYTE	7615
LSREAD	7671
M100	7665
M200	7646
M376	7707
READ	7654
STORE	7776

